






 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 
 
Figure 2.  Estimated logistic model (a) and corresponding standardized residuals (b) for 
the Chardonnay data. 
 



 

 
 
 
Figure 3.  Decumulated data for Chardonnay growth data. 
 
 
 
 
 



 
 

 
 
Figure 4.  Estimated two-component mixture model form (a) and associated 
standardized residuals (b) for the Chardonnay data. 



 
 
Figure 5.  Point of component equivalency (large dash line) and actual model minimum 
(small dotted line) for the Chardonnay data. 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
Figure 6.  Point of numerical minimum for the Chardonnay data. 



Appendix 1a.  Estimation of simple logistic model. 

 

/**********************************************************/ 

/**  NLMIXED used to get ML estimates of simple                                    **/ 

/**  logistic model.  Lc is cumulated growth data, DOY is day of year.   **/ 

/**********************************************************/ 

 

proc nlmixed data=gallo1; 

 parms m=700 k=1 l=200 sigma=20; 

 yhat = m/(1 + exp(-k*(doy-l))); 

 model lc ~normal(yhat, sigma); 

 predict  yhat out=pred; 

 predict lc - yhat out=resid; 

run; 

 

Appendix 1b.  Estimation of two component mixture model. 

 

/*********************************************************/ 

/**  NLMIXED used to get ML estimates of two- component mixture  **/ 

/**  model based on Normal PDFs. Diff is differenced cumulative         **/ 

/**   growth data and dd are degree day measurements.                         **/ 

/**********************************************************/ 

 

proc nlmixed data=gallo1; 

 parms p=.75 mu1=565 mu2=1143 sigma1=151 sigma2=127 C=6200 sigma=4; 

 yhat=C*(p*PDF("NORMAL",dd,mu1,sigma1)+(1-p)*PDF("NORMAL",dd,mu2,sigma2)); 

 model diff ~normal(yhat, sigma);  

 predict yhat out=pred; 

 predict diff-yhat out=resid; 

run; 



Appendix 2. SAS Bootstrap program for identifying percentile intervals on two component minimum. 
 

/*************************************************************/ 

/****  Define bootstrap macro.                                                                           ***/ 

/*************************************************************/ 

%MACRO BOOT(ITER); 

 

 /*************************************************************/ 

 /****  Run bootstrap loop ITER times                                                               ***/ 

 /*************************************************************/ 

 

 %DO I = 1 %TO &ITER ; 

 /******************************************************************/ 

 /****  Use SURVETSELECT to grab random sample with replacement.                  ***/ 

 /******************************************************************/ 

 

  proc surveyselect method=urs n=642 outhits data=resid out=sample; 

   id pred; 

 

 /*************************************************************/ 

 /****  Name change and add residuals to model                                           ***/ 

 /*************************************************************/ 

  data sample (keep = resid);  

   set sample; 

   resid = pred; 

  data sample (keep = boot dd yhat resid diff); 

   merge pred sample; 

   boot = yhat + resid; 

 

 /*************************************************************/ 

 /****  Fit model to bootstrap sample and save results.                                ***/ 

 /*************************************************************/ 

  ods output ParameterEstimates=ests; 

  proc nlmixed data=sample; 

   parms p=.75 mu1=565 mu2=1143 sigma1=151 sigma2=127 C=6200 sigma=4; 

   mu=C*(p*PDF("NORMAL",dd,mu1,sigma1)+ 

            (1-p)*PDF("NORMAL",dd,mu2,sigma2)); 

   model boot ~normal(mu, sigma);  

 

 /*************************************************************/ 

 /****  Numerically find minimum.  We know where to look.                        ***/ 

 /*************************************************************/ 

  proc transpose data=ests out=ests; 

   var estimate; 

   id parameter; 

  data ests; 

   set ests; 

   do dd=850 to 910 by .001; 

    mu=C*(p*PDF("NORMAL",dd,mu1,sigma1)+ 

            (1-p)*PDF("NORMAL",dd,mu2,sigma2)); 

    output; 

   end; 

  data ests; 

   set ests; 

   f1 = mu-lag1(mu); 

   if f1=. then delete; 

   test=abs(f1); 



 

  proc sort data=ests; 

   by test; 

  data ests; 

   set ests; 

   if _n_=1; 

  

 /********************************************************************/ 

 /****  Add the current minima to theset of all bootstrap results                                ***/ 

 /********************************************************************/ 

 data boots (keep=iter pred dd); 

   set boots ests; 

   iter=&ITER; 

 %END; 

%MEND; 

 

/*************************************************************/ 

/****  Fit Original data using two component distribution                           ***/ 

/*************************************************************/ 

proc nlmixed data=gallo1; 

 parms p=.75 mu1=565 mu2=1143 sigma1=151 sigma2=127 C=6200 sigma=4; 

 mu=C*(p*PDF("NORMAL",dd,mu1,sigma1)+ (1-p)*PDF("NORMAL",dd,mu2,sigma2)); 

 model diff ~normal(mu, sigma);  

 predict mu out=pred; 

 predict diff-mu out=resid; 

 

/******************************************************************/ 

/****  Quick and dirty variable name change to avoid conflicts.                             ***/ 

/******************************************************************/ 

 

data pred (drop = pred); 

 set pred; 

 yhat = pred; 

 

/*****************************************************************/ 

/****  Start a dataset for all bootstrap results.                                                        ***/ 

/*****************************************************************/ 

data boots; 

 iter=0; 

 

/*************************************************************/ 

/****  Turn off listing and start bootstrap macro.                                          ***/ 

/*************************************************************/ 

ods listing close; 

%BOOT(1000); 

run; 

 

/*************************************************************/ 

/****  Turn listing back on.  Use UNIVARIATE, SUMMARY or                       ***/ 

/****  INSIGHT on dataset BOOTS to find percentiles.                                    ***/ 

/*************************************************************/ 

ods listing; 

